PRACTICA 4. SUCESIONES NUMÉRICAS. CÁLCULO DIFERENCIAL Y CÁLCULO INTEGRAL EN UNA VARIABLE.

1 INTRODUCCIÓN.

El alumno debe ir realizando todos los EJEMPLOS que aparecen en esta práctica e ir anotando los resultados en la hoja que se le ha entregado. Además de estos ejemplos, hay planteados unos EJERCICIOS, que, a diferencia de lo que ocurre con los EJEMPLOS, no vienen desarrollados en el lenguaje Maxima, sino que tendrá que ser el alumno el que tenga que introducirlos para obtener el resultado.

Por ejemplo, si en la práctica aparece

```
-->  3+5;
```

el alumno tendrá que situarse en la línea anterior, y darle a INTRO del teclado numérico para obtener el resultado.

Sin embargo, si aparece

EJERCICIO 1: Resolver la siguiente operación:
3+5

tendrá que ser el alumno el que introduzca el resultado (por ejemplo a continuación de esta línea (pinchar con el ratón debajo de esta línea y darle a ENTER para que aparezca la flecha en rojo donde introducir la sentencia:

2 SUCESIONES DE NÚMEROS REALES.

En Maxima, las sucesiones se definen a través de su término general o bien mediante una expresión de recurrencia. En cualquier caso, se utilizará el operador :=, introduciendo la variable índice entre corchetes [].

Por ejemplo, podemos introducir la sucesión de término general a[n]:=1/(2*n-1); podemos calcular su primer término o los diez primeros para lo cual tendremos que hacer sucesivamente:

```
-->  /*EJEMPLO 1*/
     a[n]:=1/(2*n-1);
     a[1];
     makelist(a[k],k,1,10);
```

También podemos definir sucesiones por recurrencia. En este caso deberemos, además, utilizar el operador : para asignar valores a los primeros elementos, como hacemos a continuación

```
--> /*EJEMPLO 2*/
    b[1]:1;
    b[n]:=2*b[n-1];
    makelist(b[k],k,1,10);
```

Para Calcular límites usamos la función limit, que está disponible en el menú Análisis->Calcular límite... de wxMaxima. Como parámetros de esta función, tenemos que teclear la expresión de una sucesión, el nombre de la variable respecto de la cual deseamos calcular el límite y el valor hacia el que ésta tiende. En wxMaxima existen dos variables especiales, inf y minf, que representan respectivamente a +infinito y -infinito.

```
--> /*EJEMPLO 3*/
    c[n]:=(n-sqrt(n^2-4))/(n+1);
    limit(c[n],n,inf);
```

EJERCICIO 1: Calcular los siguientes límites:
a) limit(n/sqrt(n),n,inf).

b) limit(1/(sqrt(n)-sqrt(n+1)),n,inf).

c) limit(((n-2)/(n-3))^(2*n),n,inf).

d) limit((-1)^n,n,inf).

e) limit(n*(-1)^n,n,inf).

Como se puede observar, ante el cálculo del límite de (-1)^n, Maxima devuelve la palabra clave "ind", que se puede interpretar como indefinido pero acotado, mientras que ante el límite n*(-1)^n responde "und" (de undefined) para indicar que el límite es indefinido y no necesariamente acotado.

En algunas ocasiones, Maxima no tendrá suficiente información para realizar el cálculo requerido y nos solicitará información adicional, que tendremos que teclear:

```
--> /*EJEMPLO 4*/
    limit(a*n^2,n,inf);
```

y nos dirá:
Is a positive, negative, or zero?

Le introducimos entonces lo que queramos, por ejemplo

negative;

y obtendremos

-infinito

3 CÁLCULO DIFERENCIAL EN UNA VARIABLE.

3.1 ESTUDIO DE UNA FUNCIÓN REAL DE VARIABLE REAL

Veamos con un ejemplo el estudio completo de una función real de variable real. Para ello, consideremos la función
f(x)=1/(1+x^2), si x<=1; f(x)=1+log(x), si x>1
y veremos como podemos hacerle un estudio local completo:

```
--> /*EJEMPLO 5*/
    /*EJEMPLO 5.1: En primer lugar veremos como estudiar su
    dominio, puntos de corte y asíntotas*/
    g(x):=1/(1+x^2);
    h(x):=1+log(x);
```

Sabemos que el dominio de g(x) es todo R, puesto que el denominador no se anula nunca, aunque si hubiese duda resolvemos la ecuación que aparece en su denominador

```
--> solve(1+x^2=0);
```

Por otra parte, la función h(x) solamente está bien definida para aquellos valores de x para los que tiene sentido log(x), es decir, el dominio de h está formado por todos los números reales positivos. Por tanto el dominio de f(x) son todos los números reales.

```
--> /*EJEMPLO 5.1: continuación*/
    /*Para estudiar los puntos de corte con el eje de las abscisas,
    planteamos las siguientes ecuaciones*/
    solve(g(x)=0,x);
    solve(h(x)=0,x);
```

Para la ecuación g(x) = 0 no existen soluciones lo que significa que no hay puntos de corte con el eje x para x<=1. En el caso de la ecuación h(x) =0, obtenemos la solución x = e^(-1), pero este valor está fuera del intervalo de definición (donde se aplica h(x) en la función f(x)), con lo que concluimos que no existe ningún punto de corte de la gráfica de la función f(x) con el eje x.

En cuanto a posibles puntos de corte con el eje vertical, cuando x = 0 nuestra función toma el valor f(0) = g(0) = 1:

```
--> g(0);
```

En definitiva, el único punto de corte de f(x) con los ejes es el (0,1).

Para estudiar las asíntotas verticales, tendríamos que analizar si existe algún punto a en el que el límite de f(x) por la derecha o por la izquierda sea + ó - infinito, lo que ocurre típicamente en funciones racionales en las que se anula el denominador, en funciones logarítmicas, etc... En nuestro caso, la función g(x) no tiene ninguna asíntota vertical, porque su denominador es siempre distinto de cero. La función h(x) tendría una asíntota vertical en x=0, debido al logaritmo y teniendo en cuenta el dominio del logaritmo, sólo tiene sentido calcular el límite cuando x tiende a cero por la derecha:

```
 -->  /*EJEMPLO 5.1: continuación*/
      limit(h(x),x,0,plus);
```

Pero esto no afecta a f(x), ya que en los alrededores de x = 0 no toma los valores de h(x), sino de g(x), con lo que f(x) no tiene ninguna asíntota vertical.

Con respecto a asíntotas horizontales, tendremos que estudiar límites de f(x)cuando x->-infinito (en cuyo caso f(x)=g(x)) y cuando x->+inf (en cuyo caso f(x)=h(x)),

```
 -->  /*EJEMPLO 5.1: continuación*/
      limit(g(x),x,minf);
      limit(h(x),x,inf);
```

Por lo tanto, podemos concluir que f(x) no tiene ninguna asíntota vertical y sí tiene una asíntota horizontal (la recta y = 0) cuando x->-infinito.

Por lo tanto, podemos concluir que f(x) no tiene ninguna asíntota vertical y sí tiene una asíntota horizontal (la recta y = 0) cuando x->-infinito.

También podemos estudiar la continuidad y derivabilidad de f(x):
Las funciones g(x) y h(x) son continuas dentro de sus respectivos dominios, por lo tanto f es continua salvo, eventualmente, en x = 1, que es el punto que divide las regiones donde f(x) toma los valores de g(x) y de h(x). Pero los límites laterales de f(x) en este punto son distintos, pues:

```
 -->  /*EJEMPLO 5.2: Estudio de la continuidad y derivabilidad*/
      limit(g(x),x,1,minus);
      limit(h(x),x,1,plus);
```

en consecuencia, f(x) no es continua en x = 1 (tendrá una discontinuidad de salto finito).

Además, f(x) no es derivable en x = 1 (por no ser continua) pero sí en el resto de R (pues tanto g(x) como h(x) lo son para valores de x que estén dentro de sus respectivos dominios). Puesto que

```
 -->  diff(g(x),x);
      diff(h(x),x);
```

Por tanto la función derivada de f(x) es: f'(x)=-2x/(x^2+1)^2, si x<1; f'(x)=1/x, si x>1

NOTA: En este último ejemplo hemos aprendido como se calcula la derivada de una función. Ésto también puede hacerse usando los comandos que nos encontramos en Análisis->Derivar...

También podemos estudiar el crecimiento y/o la existencia de extremos relativos:

El crecimiento de f(x) depende del signo de la derivada, f'(x).
Cuando x > 1, obviamente, 1/x > 0 y por lo tanto, f'(x) > 0.
Aunque es un caso tan sencillo que no merece la pena recurrir al ordenador, puede servir como ilustración la forma en que se podría comprobar lo anterior, utilizando los comandos assume e is:

```
-->  /*EJEMPLO 5.3: Estudio de la monotonía y extremos*/
     assume(x>1);
     is(1/x>0);
```

cuando x < 1, el asunto es diferente, pues el signo de f'(x) dependerá de la expresión -2x/(x^2+1)^2, que depende del signo de -2x (puesto que el denominador es siempre estrictamente positivo), siendo, negativo cuando x > 0 y positivo en caso contrario.
Por lo tanto f(x) es creciente en (-inf,0) y (1,+inf) y decreciente en (0,1).

Aunque en casos tan sencillos como este no merezca la pena, el signo de -2x/(x^2+1)^2 se podría haber estudiado utilizando el ordenador. Aunque ante una pregunta inicial, Maxima se muestra incapaz de ofrecer una respuesta:

```
-->  forget(x>1)$
     assume(x<1)$
     expresion:diff(g(x),x);
     is(expresion>0);
```

NOTA: Es necesario utilizar el comando forget, que elimina la restricción que fue impuesta anteriormente por assume.

Evidentemente, lo que está ocurriendo es que el signo de la expresión anterior depende de -2x, es decir, de si x > 0 ó x < 0. Lo podemos comprobar:

```
-->  forget(x<1)$
     assume(0<x,x<1)$
     is(expresion<0);
```

```
-->  forget(0<x,x<1)$
     assume(x<0)$
     is(expresion>0);
```

```
-->  forget(x<0)$
```

Para hallar máximos y mínimos relativos, calculemos los puntos críticos de f:

```
--> solve( diff(g(x),x)=0, x);
```

```
--> solve( diff(h(x),x)=0, x);
```

Esto es, h no tiene puntos críticos y g tiene un punto crítico, x = 0. Cuando x = 0 la función f es igual a g, luego éste es un punto crítico de f. Para saber si se trata de un máximo o un mínimo, podemos estudiar el signo de la derivada segunda (notemos como se calcula ésta):

```
--> diff(g(x),x,2);
```

```
--> %,x=0;
```

Por lo tanto, f''(0) = g''(0) = -2 < 0, es decir, f tiene un máximo relativo en x = 0.

Finalizaremos el estudio local, haciendo una representación gráfica de la función: Para ello, podemos definir:

```
--> /*EJEMPLO 5.4: Representación gráfica*/
    f(x):= if(x<=1) then g(x) else h(x);
```

```
--> plot2d(f(x),[x,-5,5]);
```

NOTA: La fórmula empleada para la representación gráfica de f(x) (ver figura anterior) tiene un problema: erróneamente, se representa una línea vertical en x = 1 que oculta la existencia de una discontinuidad de salto en este punto.

EJERCICIO 2:
a) Estudiar la monotonía y extremos de la curva y = x^4.exp(-x^2).

b) Estudiar la monotonía, asíntotas, extremos, concavidad, puntos de inflexión y representación gráfica de las siguientes funciones:
y = (2x^3-5x^4+14x-6)/(4x^2); y = x.exp(1/x); y = arctan(x)

3.2 POLINOMIOS DE TAYLOR

En Maxima la orden taylor(f(x),x,x0,n) muestra el desarrollo en series de Taylor hasta el término n-ésimo de la función f(x) alrededor de x0 pero no permite tratar su resultado como una expresión en x.

```
--> f(x):=sqrt(x+1);
```

```
--> taylor(f(x),x,0,2);
```

Para calcular el polinomio de Taylor como expresión se utiliza la función subst(), a continuación se calculan los polinomios de Taylor de orden 2, 3 y 4 así como sus respectivas gráficas para observar como se aproximan dichos polinomios a f(x) alrededor de x0 = 0:

```
-->  /*EJEMPLO 6: Cálculo de diferentes polinomios de Taylor de la
     función anterior*/
     p2(x):=subst(t=x,taylor(f(t),t,0,2));
     p2(x);
```

```
-->  p3(x):=subst(t=x,taylor(f(t),t,0,3));
     p3(x);
```

```
-->  p4(x):=subst(t=x,taylor(f(t),t,0,4));
     p4(x);
```

Antes de representar graficamente los polinomios de Taylor, se desea comprobar que efectivamente los polinomios evaluados “cerca” del valor x0 = 0 producen valores cercanos a los de la función f(x). Para ello, aproximaremos la función y los polinomios anteriores en un punto cercano a x0=0. En este caso, tomaremos x1=0.3:

```
-->  /*Calculamos la función f(x) en el punto x1=0.3 y lo
     comparamos con los valores obtenidos en los 3 polinomios
     de Taylor*/
     x1:0.3$
     f(x1);
     p2(x1);
     p3(x1);
     p4(x1);
```

```
-->  /*Los errores de aproximación vendrán dados por*/
     abs(f(x1)-p2(x1));
     abs(f(x1)-p3(x1));
     abs(f(x1)-p4(x1));
```

Observemos que si se pretende aproximar valores algo más alejados del valor donde se está realizando el desarrollo, los polinomios de Taylor no funcionan tan bien. Por ejemplo, hacemos lo mismo anterior pero en el punto x1=1.5:

```
-->  /*Calculamos la función f(x) en el punto x1=0.3 y lo
     comparamos con los valores obtenidos en los 3 polinomios
     de Taylor*/
     x2:1.5$
     f(x2);
     p2(x2);
     p3(x2);
     p4(x2);
```

```
-->  /*Los errores de aproximación vendrán dados por*/
     abs(f(x2)-p2(x2));
     abs(f(x2)-p3(x2));
     abs(f(x2)-p4(x2));
```

Finalmente representamos la función f(x) y los 3 polinomios de Taylor que hemos obtenido para aproximarla:

```
--> /*Representamos gráficamente los 3 polinomios anteriores y la
    función f(x)*/
    plot2d([f(x),p2(x),p3(x),p4(x)], [x,-2,6],[y,-1,3]);
```

EJERCICIO 3:
a) Dada la función g(x) = ln(x + 1), calcular los polinomios de Taylor de orden 2, 3 y 4 alrededor de x0=0 y representar las correspondientes gráficas para estudiar su comportamiento.

b) Se desea aproximar el número x1 = 2^(1/2). Para ello, se considera la función f(x) =(x + 3/2)^(1/2). Encuentra el polinomio de Taylor que desarrollado en un entorno de x0 = 0 aproxima a x1 con un error absoluto menor que E = 7 x 10^(-5).
Nota: Se considera como valor “exacto” de 2^(1/2) el dado por 2^(1/2) = 1,414213562373095.

c) Sea la función p(x) = 1 + x + x^2 ¿Cuál es su polinomio de Taylor de orden 2 alrededor de x0 = 0? ¿y de x0 = 10?, ¿Será cierto para cualquier polinomio de grado n?

4 CÁLCULO INTEGRAL EN UNA VARIABLE.

4.1 CÁLCULO DE PRIMITIVAS

En wxMaxima es posible calcular integrales de funciones mediante la orden integrate(f(x),x) o mediante el botón [Integrar. . .] e introduciendo en la ventana de diálogo la correspondiente función.

```
--> /*EJEMPLO 7: Cálcular la primitiva de la siguiente función*/
    f(x):=3*x^2-x/3+1;
    integrate(f(x),x);
```

De la misma forma que ocurre con la orden taylor(), la orden integrate() no genera una expresión en x que nos permita realizar cálculos. De nuevo se utiliza la orden subst():

```
--> F(x):=subst(t=x,integrate(f(t),t));
    F(x);
```

El conjunto formado por todas las primitivas de la función f(x) viene dado por:

```
--> F(x):=subst(t=x,integrate(f(t),t))+C;
    F(x);
```

Podemos comprobar que F(x) es primitiva de f(x) derivando respecto de x:

```
--> diff(F(x),x);
```

4.2 INTEGRAL DEFINIDA

Sabemos que define el área del recinto formado por las rectas x = a, x = b, y = 0 y la gráfica de f(x) como la integral definida de la función en valor absoluto entre los extremos a y b.

Comenzaremos calculando el área bajo la curva f(x) = e^x*cos(x) entre a = 0 y b = pi:
En primer lugar nos aseguraremos si la función cambia de signo en el intervalo indicado, pues si así fuese se debe definir abs(f(x)) como una función a trozos en intervalos, y por lo tanto la integral en dicho intervalo se divide en suma de integrales.

```
-->  /*EJEMPLO 8: Cálculo de un área de un recinto plano*/
     kill(f,g,h)$
     f(x):=%e^x*cos(x)$
     solve(f(x)=0,x);
```

El mensaje de warning indica que para resolver la ecuación se han utilizado inversas de funciones trigomométricas que unicamente están definidas en ciertos intervalos y al ser funciones periódicas existen soluciones que no se están mostrando. En el intervalo [0,pi] la función cos(x) se anula una única vez, x = pi/2, y la función exponencial no se anula nunca.

El siguiente paso es ver qué signo tiene la función f(x) en cada intervalo [0,pi/2] y [pi/2,pi]. Para lo cual se elige un punto del intervalo [0,pi/2] y se calcula su imagen por la función f(x);
el signo de este valor es el signo de la función para todo ese intervalo. Se repetirá el mismo procedimiento para el segundo intervalo:

```
-->  f(1),numer;
     f(2),numer;
```

Así vemos que f(x) es positiva en el primer intervalo, y negativa en el segundo. Por ello, el valor del área pedida lo obtenemos haciendo

```
-->  integrate(f(x),x,0,%pi/2)-integrate(f(x),x,%pi/2,%pi);
```

```
-->  %,numer;
```

NOTA: En wxMaxima se puede calcular la integral definida usando el botón Análisis ->Integrar, modificando en la ventana de diálogo los límites de integración.

```
--> /*EJEMPLO 9: Calcular el área comprendida entre las gráficas
    de f(x)=3x^3-x^2-10x y g(x)=-x^2+2x*/
    /*Calcularemos en primer lugar los puntos de corte
    (si los hay)*/

    kill(f,g)$
    f(x):=3*x^3-x^2-10*x$
    g(x):=-x^2+2*x$
    solve(f(x)=g(x),x);
```

```
--> /*En lugar de estudiar que gráfica está por encima en cada
    uno de los intervalos (-2,0) y (0,2) nos limitaremos a
    calcular en cada uno de estos intervalos la integral de una
    función menos la otra, y tomaremos el resultado en
    valor absoluto*/

    Area1:integrate(f(x)-g(x),x,-2,0);
    Area2:integrate(f(x)-g(x),x,0,2);
```

```
--> /*Por tanto el area total será*/

    Areacomprendida:abs(Area1)+abs(Area2);
```

En la siguiente gráfica se puede comprobar el estudio realizado:

```
--> plot2d([f(x),g(x)],[x,-3,3]);
```

4.3 APLICACIONES DE LA INTEGRAL DEFINIDA

Vamos a ver a continuación unos sencillos ejemplos con las aplicaciones más típicas de la integral definida.

- CÁLCULO DE LONGITUDES DE CURVAS:
Sabemos que la longitud de una curva dada por y=f(x) entre dos puntos x1=a y x2=b viene dada por la expresión
L = integrate((1+(f'(x))^2)^(1/2),x,a,b)

```
--> /*EJEMPLO 10: Calcular la longitud de la curva e^x desde a = 0
    hasta b = 10*/

    kill(f,g)$
    f(x):=%e^x;
    g(x):=subst(t=x,diff(f(x),x))$
```

```
--> integrate(sqrt(1+g(x)^2),x,0,10),numer;
```

- VOLUMEN DE UN SÓLIDO DE REVOLUCIÓN:
Ya conocemos las fórmulas que nos permiten calcular el volumen del sólido obtenido al girar una región plana alrededor del eje OX o del eje OY, por lo que pasaremos directamente a un ejemplo:

```
--> /*EJEMPLO 11: Calcular el volumen que se genera al girar
    alrededor del eje OX la región determinada por
    f(x)=(4-x^2)^1/2 y g(x)=(1-x^2)^1/2 entre los puntos
    a=-1 y b=1*/

    kill(f,g)$
    f(x):=sqrt(4-x^2);
    g(x):=sqrt(1-x^2);
```

```
--> %pi*integrate(f(x)^2-g(x)^2,x,-1,1);
```

4.4 Racionales y decimales

- ÁREA DE UNA SUPERFICIE DE REVOLUCIÓN:
Dada una función f(x), el área de revolución obtenida al girar la gráfica de esta función alrededor del eje OX entre x=a y x=b viene dada por
AS(OX)=2pi*integral(f(x)*(1+(f'(x))^2)^1/2) entre a y b;
mientras que si gira en torno a OY el área viene dada por
AS(OY)=2pi*integral(x*(1+(f'(x))^2)^1/2) entre a y b.

Veamos entonces un ejemplo:

```
--> /*EJEMPLO 12: Calcular el área de revolución del sólido que
    se genera al girar la curva y =(1-x^2)^1/2 alrededor del eje OX
    entre los puntos a=-1 y b=1*/

    kill(f,g)$
    f(x):=sqrt(1-x^2);
    g(x):=subst(t=x,diff(f(x),x))$
```

```
--> 2*%pi*integrate(f(x)*sqrt(1+g(x)^2),x,-1,1);
```

EJERCICIO 4:
a) Calcular el área comprendida entre las gráficas f(x)=e^x y g(x)=4-x^2.
b) Calcular la longitud del arco de curva y =x^2x+x+1 entre a=0 y b=5.
c) Calcular el área de la superficie de revolución que se forma al girar la gráfica de la función g(x) = x alrededor del eje OX entre los puntos a = 0 y b = 10. ¿Qué superfice se forma? ¿Coincide los resultado con la fórmula de la superficie de dicha figura?.
d) Calcular el volumen del sólido de revolución que se forma al girar la región acotada por las gráficas de las funciones h1(x)=5+(1-x^2)^1/2 y h2(x)=5-(1-x^2)^1/2 alrededor del eje OX entre los puntos a=-1 y b = 1. ¿Qué sólido se forma? ¿Coincide el resultado con la fórmula del volumen de dicho sólido?.